
MIMEDefang and SpamAssassin's spamc/spamd
MIMEDefang does not use SA's spamc/spamd because it uses the SpamAssassin's Perl

API (Mail::SpamAssassin).

However, since MIMEDefang uses a preforked model like spamc/spamd, MIMEDefang

should be as fast as spamc/spamd.

This means that if you are using SpamAssassin only through MIMEDefang, you can just

save RAM and CPU by shutting down spamc/spamd. For example, in RedHat Linux:

service spamassassin stop
chkconfig spamassassin off

On the other hand, if you are using SpamAssassin through MIMEDefang and you are

also using spamc/spamd for other applications, be aware that MIMEDefang uses its own

SpamAssassin objects, and does not participate in the spamd pool. Bear this in mind

when setting the parameters for the maximum spamd and MIMEDefang threads so that

you don't over-extend your system.

How to Force MIMEDefang to use spamc
(This section originally by Matthew van Eerde)

All the previous notwithstanding, it is entirely possible to hack MIMEDefang to use spamd

via spamc. It may even be to your advantage. If your MIMEDefang threads spend a lot of

time doing other things than waiting for the SpamAssassin call to return, it can save

memory to use spamc. This memory savings comes at a cost of a few extra shell calls.

Important: The following instructions assume SpamAssassin 3.1 and SpamAssassin

configuration files located in /etc/mail/spamassassin.

First, get spamd working.

Make sure spamc with no arguments works as it should. To do this you may need to tweak

/etc/mail/spamassassin/spamc.conf.

Then, add the following into /etc/mail/spamassassin/local.cf and restart

spamd:

clear_report_template
report Score:           _SCORE()_
report Required:        _REQD_
report Tests:           _TESTS(,)_
report
report _SUMMARY_

This makes the report easier to parse.

Important: The whitespace after the colons needs to be a single tab.

Add the following at the top of your /etc/mail/mimedefang-filter file, before the

detect_and_load_perl_modules(); statement:

$Features{"Spamc"} = 1;
$Features{"SpamAssassin"} = 0; # false but defined
sub create_spamc_scan_file()
{
        open(IN, "<./INPUTMSG") or return undef;
        my @msg = <IN>;
        close(IN);

        # Synthesize a "Return-Path" and "Received:" header
        my @sahdrs;
        push (@sahdrs, "Return-Path: $Sender\n");
        push (@sahdrs, split(/^/m, synthesize_received_header()));
        push (@sahdrs, gen_msgid_header()) if ($MessageID eq "NOQUEUE");

        unshift (@msg, @sahdrs);

        open(FORSPAMC, ">./FORSPAMC") or return undef;
        print FORSPAMC @msg;
        close(FORSPAMC);

        return 1;
}

In filter_end, add this code:

# Use spamc instead of spamassassin
if (
    $Features{"Spamc"} and # is there spamc?
    -s "./INPUTMSG" < 100 * 1024 and # don't scan over 100KB
    create_spamc_scan_file()
   )

  {
    my $report = `spamc -r < ./FORSPAMC`;
    unlink('./FORSPAMC');

    if ($report eq "")
    {
        # not spam! nothing to do.
    } elsif (
        # customized report template
        $report =~
        /^
        Score:          \t      ([\d\.]+?)      \n
        Required:       \t      ([\d\.]+?)      \n
        Tests:          \t      ([\w,]+?)       \n
        \n
        /x
    )
    {
        my $score = $1;
        my $required = $2;
        my $tests = $3;
        my $stars = "*" x ($score < 40 ? int($score) : 40);

        if ($score >= $required)
        {
            action_change_header(
                "X-Spam-Score",
                $stars . " (" . $score . ") " . $tests
            );

            # action_change_header doesn't handle multiple Subject headers
            action_delete_all_headers("Subject");
            action_add_header(
                "Subject",
                "[Spam] $Subject"
            );
            md_graphdefang_log('spam', $score, $RelayAddr);
        } else
        {
            # Delete any existing X-Spam-Score header
            action_delete_header("X-Spam-Score");
        }
    } else

    {
        # malformed report!  Something's weird.
        md_syslog(

            "$QueueID: malformed spamc report"
        );
    }
}

Your MIMEDefang threads should shrink considerably... mine went from ~30MB to

~11MB.

Important:You will have to keep enough spamd threads around to keep your

MIMEDefang threads happy, but there could still be a substantial net savings.

